
1 Introduction

The term Lax Pairs refers to a set of two operators that, if they exist, indicate that a
corresponding particular evolution equation,

F (x, t, u, . . . ) = 0, (1)

is integrable. They represent a pair of differential operators having a characteristic
whereby they yield a nonlinear evolution equation when they commute. The idea was
originally published by Peter Lax in a seminal paper in 1968 [Lax-68].

A Lax pair consists of the Lax operator L (which is self-adjoint2 and may depend upon x,
ux, uxx, . . . , etc. but not explicitly upon t) and the operator M that together represent a
given partial differential equation such that Lt = [M,L] = (ML− LM). Note: [M,L] =
(ML − LM) = −[LM −ML) = −[L,M ] represents the commutator of the operators
L and M . Operator M is required to have enough freedom in any unknown parameters
or functions to enable the operator Lt = [M,L] (or Lt + [L,M ] = 0) to be chosen so
that it is of degree zero, i.e. does not contain differential operator terms and is thus a
multiplicative operator. L and M can be either scalar or matrix operators.

An important characteristic of operators is that they only operate on terms to their right!

The process of finding L and M corresponding to a given equation is generally non-trivial.
Therefore, if a clue(s) is available, inverting the process by first postulating a given L

1email: graham@griffiths1.com
2A self-adjoint operator is an operator that is its own adjoint or, if a matrix, one that is Hermitian, i.e a matrix that

is equal to its own conjugate transpose.

and M and then determining which partial differential equation they correspond to, can
sometimes lead to good results. However, this may require the determination of many
trial pairs and, ultimately, may not lead to the required solution. Because the existence
of a Lax pair indicates that the corresponding evolution equation is integrable3, finding
Lax pairs is a way of discovering new integrable evolution equations. This approach will
not be pursued here, but detailed discussion can be found in [Abl-11, Abl-91, Dra-92].

In addition, if a suitable Lax pair can be found for a particular nonlinear evolutionary
equation, then it is possible that they can be used to solve the associated Cauchy problem
using a method such as the inverse scattering transform (IST) method due to Gardener,
Greene, Kruskal, and Miura [Gar-67]. However, this is generally a difficult process and
additional discussion can be found in [Abl-11, Abl-91, Inf-00, Joh-97, Pol-07].

2 Lax pair analysis

2.1 Operator form

Given a linear operator L, that may depend upon the function u(x, t), the spatial variable
x and spatial derivatives ux, uxx, · · · , etc. but not explicitly upon the temporal variable
t, such that

Lψ = λψ, ψ = ψ(x, t), (2.1a)

the idea is to find another operator M , whereby:

ψt = Mψ. (2.1b)

Now, following Ablowitz and Clarkson [Abl-91], we take the time derivative of eqn (2.1a)
and obtain

Ltψ + Lψt =λtψ + λψt

⇓
Lψt + LMψ =λtψ + λMψ

=λtψ +MLψ

(Lt + LM −ML)ψ =λtψ. (3)

Hence, in order to solve for non-trivial eigenfunctions ψ (x, t), we must have

Lt + [L,M ] = 0, (4)

where
[L,M ] := LM −ML, (5)

which will be true if and only if λt = 0. Equation (4) is known as the Lax representation
of the given PDE, also known as the Lax equation, and [L,M ] in eqn (5) represents the
commutator of the two operators L and M which form a Lax pair. Thus, we have shown
that a Lax pair, eqns (2.1a) and (2.1b), has the following properties:

a. The eigenvalues λ are independent of time, i.e. λt = 0.

b. The quantity ψt −Mψ must remain a solution of eqn (2.1a).

c. The compatibility relationship Lt + (LM −ML) = 0 must be true; this, together
with property a., implies that the operator L must be self-adjoint, i.e. L is equal to
its complex conjuate.

3By integrable we mean the equation is exactly solvable; however, this begs the question: what does exactly solvable
mean? There are many definitions in use, but we use it to include: i) solutions obtained by linearization or direct methods
[Abl-11, chap. 5]; and ii) asymptotic solutions obtained by perturbation expansion methods [Dra-92, chap. 7].

2.2 Operator usage

Because L is a composite operator which may consist of differentials, functions and
scalars, for consistency, each term must act as an operator. Differentials are operators by
definition and, therefore, non-differentials are considered to include an identity operator.
Thus, as an example, an operator could be defined as,

L = Dn
x + αu(x, t)I, (6)

where Dn
x represents the nth total derivative operator and I the identity operator; α is a

scalar and u is a function of the spatial variable x and temporal variable t. Alternatively,
eqn (6) can be written in an equivalent form as

L =
∂n

∂xn
+ αu, u = u(x, t), (7)

where the second term is assumed to include an implicit identity operator.

Now, if L operates on an ancillary function, the result is a new function. For example, if
n = 2 and the ancillary function is ψ(x, t), we obtain

Lψ =
(
D2

x + αuI
)
ψ, (8)

=
∂2ψ

∂x2
+ αuψ. (9)

However, if L operates on a second operator, the result is another operator. For example,
if the second operator is M = ψI, we obtain

LM =
(
D2

x + αuI
)

(ψI) , (10)

=
(
D2

xψI
)

+ (αuIψI) , (11)

=Dx

(
∂ψ

∂x
I + ψDx

)
+ (αuψI) , (12)

=

(
∂2ψ

∂x2
I +

∂ψ

∂x
Dx +

∂ψ

∂x
Dx + ψD2

x

)
+ (αuψI) , (13)

=
∂2ψ

∂x2
I + 2

∂ψ

∂x
Dx + ψD2

x + αuψI. (14)

The above demonstrates that an operator acting upon another operator can become quite
involved, particularly when either or both have multiple terms. Consequently, care has
to be taken to ensure all the terms are correctly acted upon.

In subsequent examples, where appropriate, we will consider the identity operator to be
implicit and drop the symbol I.

2.3 Matrix form

In 1974 Ablowitz, Kaup, Newell, and Segur [Abl-74] published a matrix formalism for
Lax pairs where they introduced a construction that avoids the need to consider higher
order Lax operators. This method is also referred to as the AKNS method. In their
analysis they introduced the following system

DxΨ =XΨ, (15)

DtΨ =TΨ, (16)

Graham W Griffiths 3 25 May 2012



where X and T correspond to operators L and M respectively, and Ψ is an auxiliary vector
function. Matices X and T will, in general, both depend upon the time independent
eigenvalue λ, and the size of Ψ will depend upon the order of L. Thus, if L is of order
2, then vector Ψ will have two elements and X and T will each be a 2× 2 matrix. The
compatibility condition for eqns (15) and (16) is

[Dt, Dx] Ψ = Dt(XΨ)−Dx(TΨ) = (DtX)Ψ−XDtΨ− (DxT )Ψ− TDxΨ = 0. (17)

This can be written succinctly as

(DtX −DxT + [X,T ])Ψ = 0, (18)

which is known as the matrix Lax equation, and where

[X,T ] := XT − TX (19)

is defined as the matrix commutator. As a consequence of geometrical considerations,
eqn (18) is also known as the zero-curvature equation. For a more detailed discussion of
this method refer to [Abl-11, Hic-12].

3 Lax pair - operator examples

3.1 The advection equation

The well known 1D advection equation is defined as

∂u

∂t
+ c

∂u

∂x
= 0, u = (x, t). (20)

Now consider the Lax pair

L =
∂2

∂x2
− uI, u = u(x, t), (21)

and

M = −c ∂
∂x
, (22)

where again L is the Lax operator and c is a constant. For this example we have included
the identity operator, I, to re-inforce the earlier discussion on operator usage.

Inserting eqns (21) and (22) into the Lax eqn (4), we obtain

∂L

∂t
= −∂u

∂t
= [M,L] =

[
c
∂

∂x
,
∂2

∂x2
− uI

]
, (23)

where the rhs of eqn (23) is equal to − (LM −ML) = (ML− LM) = [M,L]. Expanding
each element on the rhs of eqn (23) separately, and recalling that operators operate only
on terms to their right, we obtain

LM =

(
∂2

∂x2
− uI

)(
−c ∂

∂x

)
= −c ∂

3

∂x3
+ cu

∂

∂x
, (24)

ML =

(
−c ∂

∂x

)(
∂2

∂x2
− uI

)
= −c ∂

3

∂x3
+ cu

∂

∂x
+ c

∂u

∂x
, (25)

⇓

(LM −ML) = −c∂u
∂x
. (26)

On negating eqn (23) and rearranging we arrive at

∂u

∂t
+ c

∂u

∂x
= 0, (27)

which is the 1D advection equation, eqn (20). If c > 0, u advects to the right, and if
c < 0, u advects to the left. Thus, we have shown that L and M satisfy Lax eqn (4) and
that u satisfies the advection equation. It therefore follows that the advection equation
can be thought of as the compatibility condition for the Lax pair eqns (21) and (22).

Whilst this example is not particularly useful, as eqn (20) can be solved using direct
methods, it does illustrate the general Lax pair idea.

A Maple program that derives the above result is included in Listing (1).

Listing 1: Maple program that performs a Lax pair transformation on the advection equation.

# Lax pair transformation - Advection equation

restart; with(DEtools): with(PDEtools):

alias(u=u(x,t)); declare(u(x,t));

_Envdiffopdomain :=[Dx ,x];

L:=(Dx^2-u);

M:=-c*Dx;

LM:= expand(mult(L,M));ML:= expand(mult(M,L));

Commutator :=LM -ML;

# Negating the Lax equation (L[t]+(LM -ML)=0) we obtain

pde_Advec :=-(diff(L,t)+Commutator)=0;

3.2 Korteweg-de Vries (KdV) equation (a)

The well known KdV equation is defined as

∂u

∂t
− 6u

∂u

∂x
+
∂u3

∂x3
= 0, u = u(x, t). (28)

Now consider the Lax pair [Pol-07]

L =
∂2

∂x2
− u (29)

and

M = −4
∂3

∂x3
+ 6u

∂

∂x
+ 3

∂u

∂x
, (30)

where L is the Lax operator and Lψ = λψ, (eqn (2.1a)), becomes the Sturm-Liouville
equation. NOTE: we now drop the explicit inclusion of identity operators for this example,
and also for examples that follow.

Inserting eqns (29) and (30) into the Lax eqn (4), we obtain

∂L

∂t
= −∂u

∂t
=

[
−4

∂3

∂x3
+ 6u

∂

∂x
+ 3

∂u

∂x
,
∂2

∂x2
− u
]
, (31)

where the rhs of eqn (31) is equal to − (LM −ML) = (ML− LM) = [M,L].

Expanding each element on the rhs of eqn (31) separately, we obtain

LM =− 4
∂5

∂x5
+ 10u

∂3

∂x3
+ 15

∂u

∂x

∂2

∂x2
+

(
12
∂u2

∂x2
− 6u2

)
∂

∂x

+

(
3
∂u3

∂x3
− 3u

∂u

∂x

)
(32)

ML =− 4
∂5

∂x5
+ 10u

∂3

∂x3
+ 15

∂u

∂x

∂2

∂x2
+

(
12
∂u2

∂x2
− 6u2

)
∂

∂x

+

(
4
∂u3

∂x3
− 9u

∂u

∂x

)
(33)

⇓

(LM −ML) =

(
−∂u

3

∂x3
+ 6u

∂u

∂x

)
. (34)

Therefore, on negating and rearranging eqn (31) we finally arrive at

∂u

∂t
− 6u

∂u

∂x
+
∂u3

∂x3
= 0, (35)

which is the familiar form of the KdV equation, eqn (28). Thus, we have shown that L
and M satisfy Lax eqn (4) and that u satisfies the KdV equation. It therefore follows
that the KdV equation can be thought of as the compatibility condition for the Lax pair
eqns (29) and (30).

A Maple program that derives the above result is included in Listing (2).

Listing 2: Maple program that performs a Lax pair transformation on the KdV equation.

# Lax pair transformation - KdV equation

restart; with(DEtools): with(PDEtools):

alias(u=u(x,t)); declare(u(x,t));

_Envdiffopdomain :=[Dx ,x];

L:=(Dx^2-u);

M:=-4*Dx ^3+6*u*Dx+3* diff(u,x);

LM:= expand(mult(L,M));ML:= expand(mult(M,L));

Commutator :=LM -ML;

# Negating the Lax equation (L[t]+(LM -ML)=0), we obtain

pde_KdV :=-(diff(L,t)+Commutator)=0;

3.3 Korteweg-de Vries (KdV) equation (b)

We can also demonstrate the derivation of the previous KdV equation Lax pair example
from a more general Lax pair [Dra-92] - one with unknown variables.

Consider the Lax pair

L =
∂2

∂x2
− u, u = u(x, t), (36)

and

M = α
∂3

∂x3
−B(x, t)

∂

∂x
− C(x, t), (37)

where again L is the Lax operator and the variables α, B and C in M have yet to be
determined. Inserting eqns (36) and (37) into the Lax eqn (4), we obtain

∂L

∂t
= −∂u

∂t
=

[
α
∂3

∂x3
−B(x, t)

∂

∂x
− C(x, t),

∂2

∂x2
− u
]
, (38)

where the rhs of eqn (38) is equal to − (LM −ML) = (ML− LM) = [M,L]. Expanding
each element on the rhs of eqn (38) separately, we obtain

LM =α
∂5

∂x5
− (B + αu)

∂3

∂x3
−
(

2
∂B

∂x
+ C

)
∂2

∂x2

+

(
uB − ∂B2

∂x2
− 2

∂C

∂x

)
∂

∂x
+ uC − ∂C2

∂x2
(39)

ML =α
∂5

∂x5
− (B + αu)

∂3

∂x3
−
(

3α
∂u

∂x
+ C

)
∂2

∂x2

+

(
uB − 3α

∂2u

∂x2

)
∂

∂x
+ uC − α∂u

3

∂x3
+B

∂u

∂x
(40)

⇓

(LM −ML) =

(
3α
∂u

∂x
− 2

∂B

∂x

)
∂2

∂x2
+

(
3α
∂u2

∂x2
− ∂B2

∂x2
− 2

∂C

∂x

)
∂

∂x

α
∂u3

∂x3
−B∂u

∂x
− ∂C2

∂x2
. (41)

Now, for (LM −ML) to be a multiplicative operator, the following must be true

3α
∂u

∂x
− 2

∂B

∂x
= 0 (42)

3α
∂u2

∂x2
− ∂B2

∂x2
− 2

∂C

∂x
= 0. (43)

To solve for B and C we integrate eqns (42) and (43 )with respect to x. Ignoring constants
of integration, we get

3αu− 2B = 0, (44)

3α
∂u

∂x
− ∂B

∂x
− 2C = 0, (45)

which yields

B =
3

2
αu, (46)

C =
3

4
α
∂u

∂x
. (47)

Substituting B and C into eqn (41), we obtain

(LM −ML) = −3

2
αu

∂u

∂x
+

1

4
α
∂u3

∂x3
. (48)

Thus, on negating eqn (38) and setting α = −4, we finally arrive at

∂u

∂t
− 6u

∂u

∂x
+
∂u3

∂x3
= 0, (49)

which again, is the familiar form of the KdV equation, eqn (28).

A Maple program that derives the above result is included in Listing (3).

Listing 3: Maple program that performs a Lax pair transformation on the KdV equation.

# Lax Pair transformation - KdV Equation

restart;with(DEtools): with(PDEtools):

alias(u=u(x,t),B=B(x,t),C=C(x,t));# assume(alpha=const);

declare(u,B,C);

_Envdiffopdomain :=[Dx ,x];

L:=Dx^2-u;

M:= alpha*Dx^3-B*Dx -C;

LM:= expand(mult(L,M));ML:= expand(mult(M,L));

Commutator := collect(simplify(LM -ML),Dx);

# For Commutator to be multiplicative , the following must be true where ,

eqn1:= 3*alpha*ux -2*Bx=0;

eqn2:= 3*alpha*uxx -Bxx -2*Cx=0;

#Integrate eqn1 and eqn2 ignoring integration constants

eqn3 :=3* alpha*u-2*B=0;

eqn4 :=3* alpha*ux-Bx -2*C=0;

sol1:=solve ({eqn1 ,eqn3 ,eqn4},{B,Bx,C});

# Substitute sol1 into the Lax equation (L[t]+(LM -ML)=0)

LaxEqn :=eval(diff(L,t)+subs(B=(3/2)*alpha*u,diff(B,x)=(3/2)*alpha*diff(u,x),

C=3/4* alpha*diff(u,x),Commutator))=0;

# Negating LaxEqn and setting alpha=-4, yields the KdV equation.

-subs(alpha=-4,LaxEqn);

3.4 The modified KdV (mKdV) equation

We have seen previously, when discussing the Bäcklund transformation, how application
of the Miura transformation [Miu-68a, Miu-68b] given below

u =
∂v

∂x
+ v2, u = u(x, t), v = v(x, t), (50)

converts the KdV equation into the modified KdV equation (mKdV)

∂v

∂t
− 6v2 ∂v

∂x
+
∂3v

∂x3
= 0. (51)

This suggests that it may be possible to use the Miura transformation to convert the
KdV Lax pair into a mKdV Lax pair. In fact, this does turn out to be a good approach.

The Miura transformation, when applied to the Lax pair for the KdV equation, given by
eqns (29) and (30), yields the following Lax pair for the mKdV equation

L =
∂2

∂2x
−
(
∂v

∂x
+ v2

)
, (52)

M =− 4
∂3

∂x3
+ 6

(
∂v

∂x
+ v2

)
∂

∂x
+ 3

∂

∂x

(
∂v

∂x
+ v2

)
. (53)

Inserting eqns (52) and (53) into the Lax eqn (4) and negating, we obtain

∂2v

∂x∂t
+ 2v

∂v

∂t
+
∂4v

∂x4
+ 2v

∂3v

∂x3
− 6v2 ∂

2v

∂x2
− 12v

(
∂v

∂x

)2

− 12v3 ∂v

∂x
= 0, (54)

which can be simplified to,(
2v +

∂

∂x

)(
∂v

∂t
− 6v2 ∂v

∂x
+
∂3v

∂x3

)
= 0. (55)

Clearly, the second bracketed term of eqn (55) must be equal to zero, from which we
deduce that v must be a solution to the mKdV equation, eqn (51). Thus, we have shown
that L and M satisfy Lax eqn (4) and that v satisfies the mKdV equation. It therefore
follows that the mKdV equation can be thought of as the compatibility condition for the
Lax pair eqns (52) and (53).

A Maple program that derives the above result is included in Listing (4).

Listing 4: Maple program that performs a Lax pair transformation on the mKdV equation.

# Lax pair transformation - mKdV equation

restart; with(DEtools): with(PDEtools):

alias(u=u(x,t),v=v(x,t)); declare(u,v);

_Envdiffopdomain :=[Dx ,x];

L_KdV :=(Dx^2-u); M_KdV :=-4*Dx ^3+6*u*Dx+3* diff(u,x);

# Miura transformation

tr:=u=diff(v,x) + v^2;

L_mKdV :=subs(tr,L_KdV);

M_mKdV :=subs(tr,M_KdV);

LM:= expand(mult(L_mKdV ,M_mKdV));ML:= expand(mult(M_mKdV ,L_mKdV));

Commutator :=LM -ML;

# Negating the Lax equation (L[t]+(LM -ML)=0), we obtain

LaxEqn :=-(diff(L_mKdV ,t)+Commutator)=0;

pde_mKdV :=diff(v,t) -6*v^2* diff(v,x)+diff(v,x,x,x);

LaxEqn2 := expand (2*v*pde_mKdV+diff(pde_mKdV ,x));

LE_Chk := simplify(lhs(LaxEqn)-LaxEqn2 ,size);

3.5 The Swada-Kotera (SK) equation

The Sawada-Kotera equation is defined as [Swa-74]

∂u

∂t
+
γ2u2

5

∂u

∂x
+ γu

∂3u

∂x3
+ γ

∂u

∂x

∂2u

∂x2
+
∂5u

∂x5
= 0, u = u(x, t), (56)

where a value of γ = 15 is commonly used.

Now consider the Lax pair

L =
∂3

∂x2
+

1

5
γu

∂

∂x
, u = u(x, t), (57)

and

M = 9
∂5

∂x5
+ 3γu

∂3

∂x3
+ 3γ

∂u

∂x

∂2

∂x2
+

(
1

5
γ2u2 + 2γ

∂2u

∂x2

)
∂

∂x
+ a(t), (58)

where again L is the Lax operator and c is a constant. Inserting eqns (57) and (58) into
the Lax eqn (4) and rearranging, we obtain

γ

5

∂

∂x

(
∂u

∂t
+
γ2u2

5

∂u

∂x
+ γu

∂u3

∂x3
+ γ

∂u

∂x

∂2u

∂x2
+
∂5u

∂x5

)
= 0. (59)

Clearly, the bracketed term of eqn (59) must be equal to zero, from which we deduce
that u must be a solution to the SK equation, eqn (56). Thus, we have shown that L
and M satisfy Lax eqn (4) and that u satisfies the SK equation. It therefore follows that
the SK equation can be thought of as the compatibility condition for the Lax pair eqns
(57) and (58).

A Maple program that derives the above result is included in Listing (5).

Listing 5: Maple program that performs a Lax pair transformation on the Sawada-Kotera equation.

# Lax Pair - Sawada -Kotera Equation

restart;with(DEtools):with(PDEtools):

alias(u=u(x,t)):declare(u);

_Envdiffopdomain :=[Dx ,x];

L:=Dx ^3+(1/5)*gamma*u(x,t)*Dx;

M:= 9*Dx ^5+3* gamma*u(x,t)*Dx ^3+3* gamma*diff(u(x,t),x)*Dx^2

+((1/5)*gamma ^2*u(x,t)^2+2* gamma*diff(u(x,t),x,x))*Dx + a(t);

LM:= expand(mult(L,M));

ML:= expand(mult(M,L));

Commutator := simplify(ML -LM);

# Therefore from Lax1 we get:

sol:=diff(L,t)-Commutator =0;

# Cancelling common terms gives

SW_Eqn := simplify(sol *5/( gamma*Dx),size);

4 Lax pair - matrix examples

4.1 Korteweg-de Vries (KdV) equation

We now demonstate the application of the matrix Lax pair method to the KdV equation

∂u

∂t
+ αu

∂u

∂x
+
∂u3

∂x3
= 0, u = u(x, t). (60)

Consider the matrix Lax pair

X =

[
0 1

λ− α

6
u 0

]
(61)

and

T =

 α

6

∂u

∂x
−4λ− α

3
u

−4λ2 +
λα

3
u+

α2

18
u2 +

α

6

∂2u

∂x2
−α

6

∂u

∂x

 , (62)

where X is the matrix equivalent of the Lax operator L, and T is the matrix equivalent
of the operator M .

Inserting eqns (61) and (62) into the matrix Lax equation (18) yields

cc =

 0 0

−α
6

(
∂u

∂t
+ αu

∂u

∂x
+
∂u3

∂3x

)
0

 =

[
0 0
0 0

]
. (63)

Therefore, the compatibility condition for X and T must be

∂u

∂t
+ αu

∂u

∂x
+
∂u3

∂3x
= 0. (64)

Letting α = −6 yields the familiar KdV equation (60).

A Maple program that derives the above result is included in Listing (6).

Listing 6: Maple program that performs a Lax pair matrix transformation on the KdV equation.

# Matrix Lax pair transformation - KdV equation

with(PDEtools): with(LinearAlgebra):

alias(u=u(x,t)): declare(u(x,t));

X:= Matrix ([[ 0, 1], [lambda -(1/6)*alpha*u ,0]]);

T:= Matrix ([[(1/6)*alpha*diff(u, x), -4*lambda -(1/3)*alpha*u],

[-4* lambda ^2+(1/3)*lambda*alpha*u+(1/18)*alpha ^2*u^2+(1/6)*alpha*diff(u,x,x),

-(1/6)*alpha*diff(u,x)]]);

# Lax equation - the "Compatibility Condition"

CC:= simplify(map(diff ,X,t)-map(diff ,T,x)+(X.T-T.X))=Matrix (2);

# Therefore , for compatibility , we must have:

pde_KdV :=lhs(CC)(2,1)*(-6/ alpha)=0;

4.2 Fifth order Korteweg-de Vries equation (KdV5)

We now demonstrate the application of the matrix Lax pair method to the fifth order
KdV equation

∂u

∂t
+ 30u2∂u

∂x
+ 20

∂u

∂x

∂2u

∂x2
+ 10u

∂3u

∂x3
+
∂5u

∂x5
= 0, u = u(x, t). (65)

Consider the matrix Lax pair

X =

 0
1

λ2

−γλ
2

10
u 0

 (66)

and

T =


γ

500

(
30γu

∂u

∂x
+ 50

∂3u

∂x3

)
− γ

50λ2

(
3γu2 + 10

∂2u

∂x2

)
γλ2

500

(
3γ2u3 + 30γ

(
∂u

∂x

)2

+ 40γ
∂2u

∂x2
+ 50

∂4u

∂x4

)
− γ

500

(
30γu

∂u

∂x
+ 50

∂3u

∂x3

)
 ,

(67)
where X is the matrix equivalent of the Lax operator L, and T is the matrix equivalent
of the operator M .

Inserting eqns (66) and (67) into the matrix Lax equation (18) yields

cc =

 0 0

−γλ
2

100

(
10
∂u

∂t
+ 3γ2u2∂u

∂x
+ 20γ

∂u

∂x

∂2u

∂x2
+ 10γu

∂3u

∂x3
+ 10

∂5u

∂x5

)
0

 =

[
0 0
0 0

]
.

(68)
Therefore, the compatibility condition for X and T must be

−γλ
2

100

(
10
∂u

∂t
+ 3γ2u2∂u

∂x
+ 20γ

∂u

∂x

∂2u

∂x2
+ 10γu

∂3u

∂x3
+ 10

∂5u

∂x5

)
= 0. (69)

On letting γ = 10 and simplifying eqn (69) yields the KdV5 equation (65).

A Maple program that derives the above result is included in Listing (7).

Listing 7: Maple program that performs a Lax pair matrix transformation on the 5th order KdV equation.

# KdV5 equation - Matrix Lax pair

with(PDEtools): with(LinearAlgebra):

alias(u=u(x,t)): declare(u(x,t));

X:= Matrix ([[ 0, 1/ lambda ^2], [ -(1/10)*gamma*lambda ^2*u ,0]]);

T:=( gamma /500)*Matrix ([ [10*(3* gamma*u*diff(u,x)+5* diff(u,x,x,x)),

-(10/ lambda ^2) *(3* gamma*u^2+10* diff(u,x,x))],

[lambda ^2*(3* gamma ^2*u^3+30* gamma*diff(u,x)^2+40* gamma*u*diff(u,x,x)

+50* diff(u,x,x,x,x)),

-10*(3* gamma*u*diff(u,x)+5* diff(u,x,x,x))] ]);

# Lax equation - the "Compatibility Condition"

CC:= simplify(map(diff ,X,t)-map(diff ,T,x)+(X.T-T.X))=Matrix (2);

# Therefore , for compatibility , we must have:

CC21:=lhs(CC)[2 ,1]=0;

# On letting gamma =10 and simplifying we obtain the

# KdV5 equation in cannonical form

pde_KdV5 :=subs(gamma =10,CC21)*(-1/ lambda ^2);

4.3 Sine-Gordon equation

The Sine-Gordon equation in light-cone co-ordinates4 is given by

∂2u

∂t∂x
− sin(u) = 0, u = u(x, t). (70)

A matrix Lax pair for this equation is [Lax-68]

X =

 −iλ −1

2

∂u

∂x
1

2

∂u

∂x
iλ

 , i =
√
−1, (71)

and

T =
i

4λ

[
cos(u) sin(u)
sin(u) − cos(u)

]
, (72)

where X is the matrix equivalent of the Lax operator L, and T is the matrix equivalent
of the operator M .

Inserting eqns (71) and (72) into the matrix Lax equation (18) yields

cc =

 0 − ∂2u

∂t∂x
+ sin(u)

∂2u

∂t∂x
− sin(u) 0

 =

[
0 0
0 0

]
. (73)

Therefore, the compatibility condition for X and T must be

∂2u

∂t∂x
− sin(u) = 0, (74)

i.e the Sine-Gordon equation (70).

A Maple program that derives the above result is included in Listing (8).

Listing 8: Maple program that performs a Lax pair matrix transformation on the Sine-Gordon equation.

# Sine -Gordon equation - Matrix Lax pair

with(PDEtools): with(LinearAlgebra):

alias(u=u(x,t)): declare(u(x,t));

X:= Matrix ([[ -I*lambda , -(1/2)*diff(u,x)], [(1/2)*diff(u,x),I*lambda ]]);

T:=(I/(4* lambda))*Matrix ([ [cos(u), sin(u)],

[sin(u), -cos(u)] ]);

# Lax equation - the "Compatibility Condition"

4A light cone is formed by all past and future events that can be connected by light rays.

CC:= simplify(map(diff ,X,t)-map(diff ,T,x)+(X.T-T.X))=Matrix (2);

# Therefore , for compatibility , we must have:

CC_12_21 :=(1/2) *(diff(u,x,t)) -(1/2)*sin(u)=0;

# On simplifying we obtain the Sine -Gordon equation

pde_SineG :=diff(u,x,t)-sin(u)=0;

4.4 Sinh-Gordon equation

The Sinh-Gordon equation in light-cone co-ordinates is given by

∂2u

∂t∂x
− sinh(u) = 0, u = u(x, t). (75)

A matrix Lax pair for this equation is [Qia-02]

X =

 −iλ 1

2

∂u

∂x
1

2

∂u

∂x
iλ

 , i =
√
−1, (76)

and

T =
i

4λ

[
cosh(u) − sinh(u)
sinh(u) − cosh(u)

]
, (77)

where X is the matrix equivalent of the Lax operator L, and T is the matrix equivalent
of the operator M .

Inserting eqns (76) and (77) into the matrix Lax equation (18) yields

cc =

 0
∂2u

∂t∂x
− sinh(u)

∂2u

∂t∂x
− sinh(u) 0

 =

[
0 0
0 0

]
. (78)

Therefore, the compatibility condition for X and T must be

∂2u

∂t∂x
− sinh(u) = 0, (79)

i.e the Sinh-Gordon equation (75).

A Maple program that derives the above result is included in Listing (9).

Listing 9: Maple program that performs a Lax pair matrix transformation on the Sinh-Gordon equation.

# Sinh -Gordon equation - Matrix Lax pair

with(PDEtools): with(LinearAlgebra):

alias(u=u(x,t)): declare(u(x,t));

X:= Matrix ([[ -I*lambda , (1/2)*diff(u,x)], [(1/2)*diff(u,x),I*lambda ]]);

T:=(I/(4* lambda))*Matrix ([ [cosh(u), -sinh(u)],

[sinh(u), -cosh(u)] ]);

# Lax equation - the "Compatibility Condition"

CC:= simplify(map(diff ,X,t)-map(diff ,T,x)+(X.T-T.X))=Matrix (2);

# Therefore , for compatibility , we must have:

CC_12_21 :=(1/2) *(diff(u,x,t)) -(1/2)*sinh(u)=0;

# On simplifying we obtain the Sinh -Gordon equation

pde_SinhG :=diff(u,x,t)-sinh(u)=0;

Graham W Griffiths 13 25 May 2012



4.5 Liouville equation

The Liouville equation in light-cone co-ordinates is given by

∂2u

∂t∂x
− exp(2u) = 0, u = u(x, t). (80)

A matrix Lax pair for this equation is [Wu-92]

X =

 ∂u

∂x
λ

λ −∂u
∂x

 (81)

and

T =

[
0

1

λ
exp(2u)

0 0

]
, (82)

where X is the matrix equivalent of the Lax operator L, and T is the matrix equivalent
of the operator M .

Inserting eqns (81) and (82) into the matrix Lax equation (18) yields

cc =

 ∂2u

∂t∂x
− exp(2u) 0

0 − ∂2u

∂t∂x
+ exp(2u)

 =

[
0 0
0 0

]
. (83)

Therefore, the compatibility condition for X and T must be

∂2u

∂t∂x
− exp(2u) = 0, (84)

i.e the Liouville equation (80).

A Maple program that derives the above result is included in Listing (10).

Listing 10: Maple program that performs a Lax pair matrix transformation on the Liouville equation.

# Liouville equation - Matrix Lax pair

with(PDEtools): with(LinearAlgebra):

alias(u=u(x,t)): declare(u(x,t));

X:= Matrix ([[ diff(u,x), lambda], [lambda , -diff(u,x)]]);

T:= Matrix ([ [0, exp(2*u)/lambda],

[0, 0] ]);

# Lax equation - the "Compatibility Condition"

CC:= simplify(map(diff ,X,t)-map(diff ,T,x)+(X.T-T.X))=Matrix (2);

# Therefore , for compatibility , we must have:

pde_LV :=lhs(CC)[1 ,1]=0;

